// File: Address.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.2 <0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly { size := extcodesize(account) }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
      return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value, string memory errorMessage) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{ value: value }(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data, string memory errorMessage) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.staticcall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.3._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.3._
     */
    function functionDelegateCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    function _verifyCallResult(bool success, bytes memory returndata, string memory errorMessage) private pure returns(bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: Context.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address payable) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}

// File: EVT.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.7.5;

import './SafeMath.sol';
import './Ownable.sol';

contract EVT is Ownable {
    using SafeMath for uint256;

    /// @notice EIP-20 token name for this token
    string public constant name = 'Evereth';

    /// @notice EIP-20 token symbol for this token
    string public constant symbol = 'EVT';

    /// @notice EIP-20 token decimals for this token
    uint8 public constant decimals = 18;

    /// @notice Total number of tokens in circulation
    uint256 public totalSupply = 0;

    uint256 public maxSupply = 2800000 ether; // 2.8 million EVT

    mapping(address => mapping(address => uint256)) internal allowances;

    mapping(address => uint256) internal balances;

    /// @notice A record of each accounts delegate
    mapping(address => address) public delegates;

    /// @notice A checkpoint for marking number of votes from a given block
    struct Checkpoint {
        uint256 fromBlock;
        uint256 votes;
    }

    /// @notice A record of votes checkpoints for each account, by index
    mapping(address => mapping(uint256 => Checkpoint)) public checkpoints;

    /// @notice The number of checkpoints for each account
    mapping(address => uint256) public numCheckpoints;

    /// @notice The EIP-712 typehash for the contract's domain
    bytes32 public constant DOMAIN_TYPEHASH = keccak256(
        'EIP712Domain(string name,address verifyingContract)'
    );

    /// @notice The EIP-712 typehash for the delegation struct used by the contract
    bytes32 public constant DELEGATION_TYPEHASH = keccak256(
        'Delegation(address delegatee,uint256 nonce,uint256 expiry)'
    );

    /// @notice A record of states for signing / validating signatures
    mapping(address => uint256) public nonces;

    /// @notice Contracts allowed to mint new tokens, can only be set by the governance
    mapping(address => bool) public minters;

    /// @notice An event thats emitted when an account changes its delegate
    event DelegateChanged(
        address indexed delegator,
        address indexed fromDelegate,
        address indexed toDelegate
    );

    /// @notice An event thats emitted when a delegate account's vote balance changes
    event DelegateVotesChanged(
        address indexed delegate,
        uint256 previousBalance,
        uint256 newBalance
    );

    /// @notice The standard EIP-20 transfer event
    event Transfer(address indexed from, address indexed to, uint256 amount);

    /// @notice The standard EIP-20 approval event
    event Approval(
        address indexed owner,
        address indexed spender,
        uint256 amount
    );

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements
     *
     * - `to` cannot be the zero address.
     */
    function _mint(address _to, uint256 _amount) internal {
        require(_to != address(0), 'EVT::_mint: mint to the zero address');

        totalSupply = totalSupply.add(_amount);
        balances[_to] = balances[_to].add(_amount);
        emit Transfer(address(0), _to, _amount);

        _moveDelegates(address(0), delegates[_to], _amount);
    }

    // Only owner (EVTMaster Contract)
    function mint(address _to, uint256 amount) public onlyOwner returns (bool) {
        require(totalSupply.add(amount, 'EVT::mint: mint amount overflows') <= maxSupply, 'EVT::mint: max supply exceeded');
        _mint(_to, amount);
        return true;
    }

    /**
     * @notice Get the number of tokens `spender` is approved to spend on behalf of `account`
     * @param account The address of the account holding the funds
     * @param spender The address of the account spending the funds
     * @return The number of tokens approved
     */
    function allowance(address account, address spender)
        external
        view
        returns (uint256)
    {
        return allowances[account][spender];
    }

    /**
     * @notice Approve `spender` to transfer up to `amount` from `src`
     * @dev This will overwrite the approval amount for `spender`
     *  and is subject to issues noted [here](https://eips.trxeum.org/EIPS/eip-20#approve)
     * @param spender The address of the account which may transfer tokens
     * @param amount The number of tokens that are approved (2^256-1 means infinite)
     * @return Whtrx or not the approval succeeded
     */
    function approve(address spender, uint256 amount)
        external
        returns (bool)
    {
        allowances[msg.sender][spender] = amount;

        emit Approval(msg.sender, spender, amount);
        return true;
    }

    /**
     * @notice Get the number of tokens held by the `account`
     * @param account The address of the account to get the balance of
     * @return The number of tokens held
     */
    function balanceOf(address account) external view returns (uint256) {
        return balances[account];
    }

    /**
     * @notice Transfer `amount` tokens from `msg.sender` to `dst`
     * @param dst The address of the destination account
     * @param amount The number of tokens to transfer
     * @return Whtrx or not the transfer succeeded
     */
    function transfer(address dst, uint256 amount) external returns (bool) {
        _transferTokens(msg.sender, dst, amount);
        return true;
    }

    /**
     * @notice Transfer `amount` tokens from `src` to `dst`
     * @param src The address of the source account
     * @param dst The address of the destination account
     * @param amount The number of tokens to transfer
     * @return Whtrx or not the transfer succeeded
     */
    function transferFrom(
        address src,
        address dst,
        uint256 amount
    ) external returns (bool) {
        address spender = msg.sender;
        uint256 spenderAllowance = allowances[src][spender];

        if (spender != src && spenderAllowance != uint256(-1)) {
            uint256 newAllowance = spenderAllowance.sub(amount);
            allowances[src][spender] = newAllowance;

            emit Approval(src, spender, newAllowance);
        }

        _transferTokens(src, dst, amount);
        return true;
    }

    /**
     * @notice Delegate votes from `msg.sender` to `delegatee`
     * @param delegatee The address to delegate votes to
     */
    function delegate(address delegatee) public {
        return _delegate(msg.sender, delegatee);
    }

    /**
     * @notice Delegates votes from signatory to `delegatee`
     * @param delegatee The address to delegate votes to
     * @param nonce The contract state required to match the signature
     * @param expiry The time at which to expire the signature
     * @param v The recovery byte of the signature
     * @param r Half of the ECDSA signature pair
     * @param s Half of the ECDSA signature pair
     */
    function delegateBySig(
        address delegatee,
        uint256 nonce,
        uint256 expiry,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) public {
        bytes32 domainSeparator = keccak256(
            abi.encode(DOMAIN_TYPEHASH, keccak256(bytes(name)), address(this))
        );
        bytes32 structHash = keccak256(
            abi.encode(DELEGATION_TYPEHASH, delegatee, nonce, expiry)
        );
        bytes32 digest = keccak256(
            abi.encodePacked('\x19\x01', domainSeparator, structHash)
        );
        address signatory = ecrecover(digest, v, r, s);
        require(
            signatory != address(0),
            'EVT::delegateBySig: invalid signature'
        );
        require(
            nonce == nonces[signatory]++,
            'EVT::delegateBySig: invalid nonce'
        );
        require(block.timestamp <= expiry, 'EVT::delegateBySig: signature expired');
        return _delegate(signatory, delegatee);
    }

    /**
     * @notice Gets the current votes balance for `account`
     * @param account The address to get votes balance
     * @return The number of current votes for `account`
     */
    function getCurrentVotes(address account) external view returns (uint256) {
        uint256 nCheckpoints = numCheckpoints[account];
        return
            nCheckpoints > 0 ? checkpoints[account][nCheckpoints - 1].votes : 0;
    }

    /**
     * @notice Determine the prior number of votes for an account as of a block number
     * @dev Block number must be a finalized block or else this function will revert to prevent misinformation.
     * @param account The address of the account to check
     * @param blockNumber The block number to get the vote balance at
     * @return The number of votes the account had as of the given block
     */
    function getPriorVotes(address account, uint256 blockNumber)
        public
        view
        returns (uint256)
    {
        require(
            blockNumber < block.number,
            'EVT::getPriorVotes: not yet determined'
        );

        uint256 nCheckpoints = numCheckpoints[account];
        if (nCheckpoints == 0) {
            return 0;
        }

        // First check most recent balance
        if (checkpoints[account][nCheckpoints - 1].fromBlock <= blockNumber) {
            return checkpoints[account][nCheckpoints - 1].votes;
        }

        // Next check implicit zero balance
        if (checkpoints[account][0].fromBlock > blockNumber) {
            return 0;
        }

        uint256 lower = 0;
        uint256 upper = nCheckpoints - 1;
        while (upper > lower) {
            uint256 center = upper - (upper - lower) / 2; // ceil, avoiding overflow
            Checkpoint memory cp = checkpoints[account][center];
            if (cp.fromBlock == blockNumber) {
                return cp.votes;
            } else if (cp.fromBlock < blockNumber) {
                lower = center;
            } else {
                upper = center - 1;
            }
        }
        return checkpoints[account][lower].votes;
    }

    function _delegate(address delegator, address delegatee) internal {
        address currentDelegate = delegates[delegator];
        uint256 delegatorBalance = balances[delegator];
        delegates[delegator] = delegatee;

        emit DelegateChanged(delegator, currentDelegate, delegatee);

        _moveDelegates(currentDelegate, delegatee, delegatorBalance);
    }

    function _transferTokens(
        address src,
        address dst,
        uint256 amount
    ) internal {
        require(
            src != address(0),
            'EVT::_transferTokens: cannot transfer from the zero address'
        );
        require(
            dst != address(0),
            'EVT::_transferTokens: cannot transfer to the zero address'
        );

        balances[src] = balances[src].sub(amount, 'EVT::_transferTokens: transfer amount exceeds balance');
        balances[dst] = balances[dst].add(amount, 'EVT::_transferTokens: transfer amount overflows');
        emit Transfer(src, dst, amount);

        _moveDelegates(delegates[src], delegates[dst], amount);
    }

    function _moveDelegates(
        address srcRep,
        address dstRep,
        uint256 amount
    ) internal {
        if (srcRep != dstRep && amount > 0) {
            if (srcRep != address(0)) {
                uint256 srcRepNum = numCheckpoints[srcRep];
                uint256 srcRepOld = srcRepNum > 0
                    ? checkpoints[srcRep][srcRepNum - 1].votes
                    : 0;
                uint256 srcRepNew = srcRepOld.add(amount, 'EVT::_moveVotes: vote amount underflows');
                _writeCheckpoint(srcRep, srcRepNum, srcRepOld, srcRepNew);
            }

            if (dstRep != address(0)) {
                uint256 dstRepNum = numCheckpoints[dstRep];
                uint256 dstRepOld = dstRepNum > 0
                    ? checkpoints[dstRep][dstRepNum - 1].votes
                    : 0;
                uint256 dstRepNew = dstRepOld.add(amount, 'EVT::_moveVotes: vote amount overflows');
                _writeCheckpoint(dstRep, dstRepNum, dstRepOld, dstRepNew);
            }
        }
    }

    function _writeCheckpoint(
        address delegatee,
        uint256 nCheckpoints,
        uint256 oldVotes,
        uint256 newVotes
    ) internal {
        uint256 blockNumber = block.number;

        if (
            nCheckpoints > 0 &&
            checkpoints[delegatee][nCheckpoints - 1].fromBlock == blockNumber
        ) {
            checkpoints[delegatee][nCheckpoints - 1].votes = newVotes;
        } else {
            checkpoints[delegatee][nCheckpoints] = Checkpoint(
                blockNumber,
                newVotes
            );
            numCheckpoints[delegatee] = nCheckpoints + 1;
        }

        emit DelegateVotesChanged(delegatee, oldVotes, newVotes);
    }
}

// File: EVTMaster.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.7.5;

import "./EVT.sol";
import "./Address.sol";

contract EVTMaster is Ownable {
    using SafeMath for uint256;
    using Address for address;

    EVT public rewardToken;
    address public lpMiningContract;

    uint256 private _lastMintTime;
    uint256 private _totalStaked = 0;
    uint256 private _platformFees = 0;
    uint256 private _mintedCurrentLevel = 0;
    uint256 private _mintedByLP = 0;
    uint256 private _mintedByStaking = 0;
    uint256 private _currentLevel = 0;

    uint256 constant MIN_INVEST = 10000000000000000;
    uint256 constant REWARD_INTERVAL = 1 days;

    uint256 constant REF_REWARD_PERCENT = 10;

    uint256 constant UNSTAKE_FEE = 10;

    uint256 constant MAX_MINT_BY_LP = 800000 ether;
    uint256 constant MAX_MINT_BY_STAKING = 2000000 ether;

    uint256[8] public levelChangeTime;

    uint256[] public LEVEL_LIMIT = [
        700000 ether,
        600000 ether,
        500000 ether,
        400000 ether,
        300000 ether,
        200000 ether,
        100000 ether,
        0
    ];

    uint256[] public LEVEL_YIELD = [
        93,
        78,
        64,
        50,
        37,
        24,
        11,
        0
    ];

    // Info of each user.
    struct User {
        uint256 investment;
        uint256 lastClaim;
        address referrer;
        uint256 referralReward;
        uint256 totalReferrals;
        address addr;
        bool exists;
    }

    mapping(address => User) private _users;

    event Operation(
        string _type,
        address indexed _user,
        address indexed _referrer,
        uint256 amount,
        uint256 timestamp
    );

    event LevelChanged(uint256 newLevel, uint256 timestamp);
    event NewReferral(
        address indexed _user,
        address referral,
        uint256 timestamp
    );
    event ReferralReward(
        address indexed _referrer,
        address indexed _user,
        uint256 amount,
        uint256 timestamp
    );
    event ClaimStaked(
        address indexed _user,
        address indexed _referrer,
        uint256 amount,
        uint256 timestamp
    );
    event ClaimReferral(address indexed _user, uint256 _amount);

    constructor(EVT _rewardToken) {
        User storage user = _users[msg.sender];
        user.exists = true;
        user.addr = msg.sender;
        user.totalReferrals = 1;
        user.referrer = msg.sender;
        user.lastClaim = block.timestamp;

        _lastMintTime = block.timestamp;
        rewardToken = _rewardToken;
    }

    function setLPContract(address _address) public onlyOwner {
        lpMiningContract = _address;
    }

    receive() external payable {
        if (msg.sender == owner() || msg.value < MIN_INVEST) {
            _refund();
        } else {
            _stake(msg.sender, address(0x0), msg.value);
        }
    }

    function stake() public payable {
        stake(address(0x0));
    }

    function stake(address _referrer) public payable {
        if (msg.sender == owner() || msg.value < MIN_INVEST) {
            _refund();
        } else {
            _stake(msg.sender, _referrer, msg.value);
        }
    }

    function _refund() private {
        _mintTokens();
        safeSendValue(msg.sender, msg.value);
    }

    function _stake(
        address _address,
        address _referrer,
        uint256 _amount
    ) private {
        require(_amount >= MIN_INVEST, 'Too low value');
        require(_address != owner(), "Owner can't stake");
        _mintTokens();

        address referrer = _referrer == address(0x0) ? owner() : _referrer;
        if (!_users[referrer].exists) {
            referrer = owner();
        }

        User storage user = _users[_address];
        if (!user.exists) {
            user.exists = true;
            user.addr = _address;
            user.referrer = referrer;
            user.investment = _amount;
            user.lastClaim = block.timestamp;

            _users[referrer].totalReferrals = _users[referrer]
                .totalReferrals
                .add(1);

            emit NewReferral(referrer, user.addr, block.timestamp);
        } else {
            _claimStaked(_address);
            user.investment = user.investment.add(_amount);
        }
        _totalStaked = _totalStaked.add(_amount);
        _platformFees = _platformFees.add(_amount.mul(UNSTAKE_FEE));
        emit Operation('stake', user.addr, user.referrer, _amount, block.timestamp);
    }

    function unstake(uint256 _amount) public {
        require(msg.sender != owner(), "Owner can't unstake");
        _mintTokens();

        User storage user = _users[msg.sender];

        require(user.exists, 'Invalid User');

        _claimStaked(msg.sender);

        emit Operation('unstake', user.addr, user.referrer, _amount, block.timestamp);

        _totalStaked = _totalStaked.sub(_amount);
        user.investment = user.investment.sub(
            _amount,
            'EVTMaster::unstake: Insufficient funds'
        );
        _amount = _amount.mul(uint256(100).sub(UNSTAKE_FEE)).div(100);

        safeSendValue(msg.sender, _amount);
    }

    function unstake() public {
        unstake(_users[msg.sender].investment);
    }

    function claimStaked() public {
        _mintTokens();
        _claimStaked(msg.sender);
    }

    function claimReferralReward() public {
        _mintTokens();
        User storage user = _users[msg.sender];
        uint256 refReward = user.referralReward;
        user.referralReward = 0;
        safeTokenTransfer(user.addr, refReward);
        emit ClaimReferral(user.addr, refReward);
    }

    function _mintTokens() internal returns (uint256 mintedAmount) {
        uint256 timePassed = block.timestamp.sub(_lastMintTime);
        if (timePassed == 0) {
            return 0;
        }
        if (_totalStaked != 0) {
            uint256 toMint = _totalStaked
                .mul(timePassed)
                .div(REWARD_INTERVAL)
                .mul(LEVEL_YIELD[_currentLevel])
                .div(100);
            // Add ref percent
            toMint = toMint.add(toMint.mul(REF_REWARD_PERCENT).div(100));
            if (_mintedByStaking.add(toMint) > MAX_MINT_BY_STAKING) {
                toMint = MAX_MINT_BY_STAKING.sub(_mintedByStaking);
            }
            mintedAmount = safeMint(address(this), toMint);
            _mintedByStaking = _mintedByStaking.add(mintedAmount);
        }
        _lastMintTime = block.timestamp;
    }

    function _claimStaked(address _address) internal {
        require(_address != owner(), "Owner can't unstake");
        User storage user = _users[_address];

        require(user.exists, 'Invalid User');

        uint256 reward = pendingReward(msg.sender);

        user.lastClaim = block.timestamp;

        uint256 referralReward = reward.mul(REF_REWARD_PERCENT).div(100);

        safeTokenTransfer(user.addr, reward);

        _users[user.referrer].referralReward = _users[user.referrer]
            .referralReward
            .add(referralReward);

        emit ClaimStaked(user.addr, user.referrer, reward, block.timestamp);
        emit ReferralReward(user.referrer, user.referrer, referralReward, block.timestamp);
    }

    function pendingReward() public view returns (uint256) {
        return pendingReward(msg.sender);
    }

    function pendingReward(address _address)
        public
        view
        returns (uint256 reward)
    {
        User memory user = _users[_address];
        uint256 lastClaim = user.lastClaim;
        for (uint256 lvl = 0; lvl <= _currentLevel; ++lvl) {
            uint256 time = (levelChangeTime[lvl] == 0)
                ? block.timestamp
                : levelChangeTime[lvl];
            if (_users[_address].lastClaim >= time) {
                continue;
            }
            reward = reward.add(
                user
                    .investment
                    .mul(time.sub(lastClaim))
                    .div(REWARD_INTERVAL)
                    .mul(LEVEL_YIELD[lvl])
                    .div(100)
            );
            if (time == block.timestamp) {
                break;
            }
            lastClaim = time;
        }
    }

    // Function for owner to withdraw staking fees
    function withdrawFees() public onlyOwner returns (uint256) {
        return withdrawFees(payable(owner()), _platformFees);
    }

    // Function for owner to withdraw staking fees
    function withdrawFees(address payable _address, uint256 _amount)
        public
        onlyOwner
        returns (uint256)
    {
        _platformFees = _platformFees.sub(_amount, 'Not enough fees!');
        return safeSendValue(_address, _amount);
    }

    // View function in order to optimize web UI
    function stats() view public returns (
        uint256 currentLevel,
        uint256 currentLevelYield,
        uint256 currentLevelSupply,
        uint256 mintedCurrentLevel,
        uint256 totalEVT,
        uint256 totalStaked,
        uint256 mintedByStaking,
        uint256 mintedByLP
    ) {
        currentLevel = _currentLevel;
        currentLevelYield = LEVEL_YIELD[_currentLevel];
        currentLevelSupply = LEVEL_LIMIT[_currentLevel];
        mintedCurrentLevel = _mintedCurrentLevel;
        mintedByStaking = _mintedByStaking;
        mintedByLP = _mintedByLP;
        totalStaked = _totalStaked;
        totalEVT = rewardToken.totalSupply();
    }

    // View function in order to optimize web UI
    function userInfo() view public returns (
        uint256, uint256, address, uint256, uint256, uint256, uint256, uint256
    ) {
        return userInfo(msg.sender);
    }

    // View function in order to optimize web UI
    function userInfo(address _address) view public returns (
        uint256 investment,
        uint256 lastClaim,
        address referrer,
        uint256 referralReward,
        uint256 totalReferrals,
        uint256 pendingRewards,
        uint256 tokenBalance,
        uint256 balance
    ) {
        investment = _users[_address].investment;
        lastClaim = _users[_address].lastClaim;
        referrer = _users[_address].referrer;
        referralReward = _users[_address].referralReward;
        totalReferrals = _users[_address].totalReferrals;
        pendingRewards = pendingReward(_address);
        tokenBalance = rewardToken.balanceOf(_address);
        balance = _address.balance;
    }

    // Don't care about rewards and stats (emergency only, in case if unexpected error occurs)
    function emergencyWithdraw(address payable _user, uint256 _amount) public returns (uint256) {
        _users[msg.sender].investment = _users[_user].investment.sub(_amount);
        _totalStaked = _totalStaked.sub(_amount);

        return safeSendValue(_user, _amount.mul(uint256(100).sub(UNSTAKE_FEE)).div(100));
    }

    // Don't care about rewards and stats (emergency only, in case if unexpected error occurs)
    function emergencyWithdraw(address payable _user) public returns (uint256) {
        return emergencyWithdraw(_user, _users[_user].investment);
    }

    // Mint EVT for LP stakers
    // This is done through EVTMaster to keep traking minting levels
    function mintFromLP(uint256 _amount) public returns (uint256 mintedAmount) {
        require(msg.sender == lpMiningContract, 'Not the LP Master');
        _mintTokens();
        if (_amount >= MAX_MINT_BY_LP.sub(_mintedByLP)) {
            _amount = MAX_MINT_BY_LP.sub(_mintedByLP);
        }
        mintedAmount = safeMint(msg.sender, _amount);
        _mintedByLP = _mintedByLP.add(mintedAmount);
    }

    // Safe token transfer function to avoid rounding errors to cause ETH being locked
    function safeTokenTransfer(address _to, uint256 _amount) internal returns (uint256) {
        uint256 balance = rewardToken.balanceOf(address(this));
        if (_amount > balance) {
            _amount = balance;
        }

        rewardToken.transfer(_to, _amount);
        return _amount;
    }

    // Safe token mint function to avoid rounding and max supply errors to cause ETH being locked
    function safeMint(address _to, uint256 _amount) internal returns (uint256) {
        uint256 canBeMinted = rewardToken.maxSupply().sub(rewardToken.totalSupply());
        if (_amount > canBeMinted) {
            _amount = canBeMinted;
        }

        _mintedCurrentLevel = _mintedCurrentLevel.add(_amount);

        rewardToken.mint(_to, _amount);

        if (
            _mintedCurrentLevel >= LEVEL_LIMIT[_currentLevel] &&
            _currentLevel < (LEVEL_LIMIT.length - 1)
        ) {
            levelChangeTime[_currentLevel] = block.timestamp;
            _currentLevel++;
            _mintedCurrentLevel = 0;
            emit LevelChanged(_currentLevel, block.timestamp);
        }

        return _amount;
    }

    function safeSendValue(address payable _to, uint256 _amount) internal returns (uint256) {
        if (_amount > address(this).balance) {
            _amount = address(this).balance;
        }
        _to.transfer(_amount);
        return _amount;
    }
}

// File: IERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}
// File: Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

import "./Context.sol";
/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor () {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(_owner == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        emit OwnershipTransferred(_owner, address(0));
        _owner = address(0);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        emit OwnershipTransferred(_owner, newOwner);
        _owner = newOwner;
    }
}

// File: SafeERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

import "./IERC20.sol";
import "./SafeMath.sol";
import "./Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using SafeMath for uint256;
    using Address for address;

    function safeTransfer(IERC20 token, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(IERC20 token, address spender, uint256 value) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        // solhint-disable-next-line max-line-length
        require((value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).add(value);
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).sub(value, "SafeERC20: decreased allowance below zero");
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) { // Return data is optional
            // solhint-disable-next-line max-line-length
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}
// File: SafeMath.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.7.5;

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, errorMessage);

        return c;
    }

    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        return add(a, b, "SafeMath: addition overflow");
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return sub(a, b, "SafeMath: subtraction overflow");
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");

        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return div(a, b, "SafeMath: division by zero");
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return mod(a, b, "SafeMath: modulo by zero");
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts with custom message when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b != 0, errorMessage);
        return a % b;
    }
}